

Глава 3

Работаем с именами

В этой главе...

- Особенности и преимущества использования имен в Excel
- Различия между именами уровня листа и книги
- Работа в диалоговом окне диспетчера имен
- Способы создания имен ячеек и диапазонов
- Создание имен, распространяющихся на множество рабочих листов
- Основные операции с именами ячеек и диапазонов
- Как Excel работает с именами ячеек и диапазонов
- Потенциальные проблемы, возникающие при использовании имен
- Секреты имен
- Примеры именованных ограничений и именованных формул

Большинство пользователей Excel знакомы с такими понятиями, как именованные ячейки и диапазоны. Работа с именованными ячейками и диапазонами дает значительные преимущества. Прочитав эту главу, вы узнаете о различных типах имен, которые могут быть использованы в Excel. Возможно, вы и не подозреваете о том, как именованные ячейки смогут помочь вам в работе.

Что такое имя

Имя можно рассматривать как идентификатор какого-либо элемента рабочей книги. Имя присваивается таким элементам, как ячейка, диапазон, диаграмма, фигура и т.д. Имя, присвоенное, например, диапазону, может затем использоваться в формулах. Предположим, что диапазон B2 : B200 рабочего листа содержит информацию о дневном обороте компании. В ячейке C1 задана ставка комиссионного сбора с продажи. В следующей формуле сумма дневного оборота умножается на ставку комиссионного сбора:

=СУММ (B2 : B200) * C1

Такая формула будет работать, однако смысл ее не очевиден. Для того чтобы формула стала понятной, следует назначить описательные имена диапазону, содержащему информацию о дневном обороте, и ячейке, в которой указана ставка комиссионного сбора.

Например, диапазону B2:B200 можно присвоить имя ДневныеПродажи, а ячейке C1 — имя СтавкаКомиссионных. Теперь формулу необходимо переписать, используя имена вместо адресов диапазона и ячейки:

=СУММ(ДневныеПродажи)*СтавкаКомиссионных

Как видите, теперь формула говорит сама за себя. Формулу, в которой вместо ссылок на ячейки используются имена, понять значительно легче.

Применение именованных ячеек и диапазонов имеет ряд преимуществ.

- Имена позволяют понять смысл формулы и использовать ее. Это особенно важно для тех, кто не участвовал в создании рабочего листа. Очевидно, что такую формулу, как, например, =Доходы-Налоги, понять значительно легче, чем =D20-D40.
- Описательное имя диапазона (например, Общий_доход) запоминается легче, чем адреса ячеек (например, A21). Кроме того, вероятность опечатки при написании имени меньше, чем при вводе адреса ячейки или диапазона.
- Вы можете легко перемещаться по рабочему листу, используя поле Имя, расположенное в левой части строки формул (щелкните на кнопке со стрелкой, и появится список всех назначенных имен). Перейти от одного элемента листа к другому можно, выбрав команду Найти и выделить ⇨ Перейти (Find&Select ⇨ Go To) или нажав клавишу <F5>. В открывшемся диалоговом окне следует ввести имя диапазона или ячейки.
- Когда вы выделяете именованную ячейку или диапазон, имя этого элемента появляется в поле Имя.
- Возможно, вам будет легче записывать формулы, используя имена ячеек и диапазонов. Эти имена вы сможете вставить в формулу, выбрав команду Использовать в формуле (Use in Formula) во вкладке Формулы и выбрав в открывшемся списке нужное имя.
- Использование имен вместо адресов ячеек и диапазонов облегчит создание макросов и дальнейшую работу с ними.

Область определения имен

Перед тем как перейти к теме создания имен и работы с ними, важно осознать, что все имена имеют свою область определения. Под *областью определения* понимается множество мест, где может использоваться имя. Областью определения имени может быть либо отдельный лист, либо вся рабочая книга.

Ссылки на имена

Вы можете ссылаться на имена уровня рабочей книги, используя это имя на любом листе рабочей книги. Ссылки на имена уровня рабочего листа должны предваряться именем соответствующего листа, если, конечно, используются на листе, отличном от того, на котором определены.

Для примера предположим, что у нас есть рабочая книга, содержащая два листа: Лист1 и Лист2. В этой рабочей книге определены следующие имена: ОбщиеПродажи (имя уровня рабочей книги), ПродажиВЕвропе (имя уровня рабочего листа Лист1) и ПродажиВАмерике (имя уровня листа Лист2). На обоих листах книги можно обратиться к имени ОбщиеПродажи следующим образом:

=ОбщиеПродажи

Если вы находитесь на листе Лист1, то к имени ПродажиВЕвропе на этом листе можно также обратиться только по имени, так как оно определено именно на этом листе:

=ПродажиВЕвропе

Однако, если на листе Лист1 вы захотите обратиться к имени ПродажиВАмерике, вам придется проделать немного больше работы. Лист1 не “видит” это имя, поскольку оно определено на другом листе. Лист1 может “видеть” только имена, определенные на нем. Для обращения к имени ПродажиВАмерике на листе Лист1 его придется предварить именем рабочего листа и восклицательным знаком:

=Лист2!ПродажиВАмерике



Совет

Если имя рабочего листа содержит пробел, заключите его в одинарные кавычки при обращении к имени, определенному на нем, например = 'Мой Лист' !МоеИмя.

В общем случае рекомендуется задавать именам как можно более узкую область определения. Если имя используется только на одном рабочем листе, ограничьте его область определения только этим листом. Если же имя используется на множестве листов рабочей книги, более подходящей будет область определения уровня книги.



Примечание

В поле Имя отображаются только имена уровня текущего рабочего листа. Аналогично, только имена уровня текущего рабочего листа отображаются в списке команды Использовать в формуле.

Ссылка на имена другой рабочей книги

В главе 2 рассказывалось о том, как создавать ссылки на ячейки и диапазоны других рабочих книг. Те же правила применяются и к ссылкам на имена, заданные в другой рабочей книге.

Например, в следующей формуле используется имя ЕжемесячныеПродажи, назначенное в рабочей книге под именем Бюджет.xls (предполагается, что эта рабочая книга открыта):

=СРЗНАЧ(Бюджет.xls!ЕжемесячныеПродажи)

Конфликты имен

Использование имен уровня рабочего листа может быть несколько запутанным, поскольку Excel разрешает на рабочих листах создавать те же имена, которые уже определены на уровне рабочей книги. В этом случае имена уровня рабочего листа имеют приоритет, но только в пределах рабочего листа, на котором определены.

К примеру, вы можете определить имя Итог уровня рабочей книги для некоторой ячейки листа Лист1. Одновременно вы можете определить имя уровня рабочего листа Лист2!Итог. Когда активен лист Лист2, Итог будет ссылаться на имя уровня рабочего листа. В то же время, если активен какой-либо другой лист, Итог будет ссылаться на имя уровня рабочей книги. Запутанно, не правда ли? Для облегчения жизни рекомендуется избегать использования одних и тех же имен на уровнях книги и листа.

Одним из способов предохранения от конфликтов этого типа является при создании имен следование некоторому соглашению об именах. При использовании соглашений имена смогут многое о себе рассказать. К примеру, во всех именах уровня рабочей книги можно использовать префикс *рк*, а в именах уровня рабочего листа — *рл*. Используя это соглашение, вы никогда не спутаете имена *рлИтог* и *ркИтог*.

Диспетчер имен

Теперь, когда вы поняли концепцию области определения имен, можно приступить к созданию и использованию имен. Excel 2007 предлагает новый способ управления именами — это диспетчер имен, окно которого показано на рис. 3.1.




Рис. 3.1. Диалоговое окно диспетчера имен

Для открытия окна диспетчера имен во вкладке **Формулы** выберите команду **Диспетчер Имен (Name Manager)** или нажмите клавиши **<Ctrl+F3>**. В этом диалоговом окне можно просматривать, создавать, редактировать и удалять имена. В главном окне диспетчера имен отображается текущее значение каждого имени, ячейка или диапазон, на которую оно ссылается, область определения имени, а также комментарии, если вы их создавали. Имена можно сортировать, а ширину столбцов изменять. Если вы используете множество имен, вы можете применить к списку некоторый предопределенный фильтр, чтобы работать только с теми именами, которые вас интересуют.

Следует обратить внимание, что размеры диалогового окна диспетчера имен можно изменять; перетаскивая за правый нижний угол, это диалоговое окно можно сделать шире или выше.

Создание имен

Диспетчер имен содержит кнопку **Создать (New)**, при щелчке на которой открывается диалоговое окно, показанное на рис. 3.2.




Рис. 3.2. Диалоговое окно создания имени

В диалоговом окне создания имени вы можете задать имя, область определения и диапазон, на который ссылается имя. При желании вы можете сопроводить имя комментарием, который поможет впоследствии понять назначение имени. Поле **Диапазон (Refers To)** позволяет выбрать ячейки с помощью мыши или ввести диапазон вручную, аналогично тому, как это выполняется в строке формул.

Редактирование имен

При щелчке на кнопке **Изменить (Edit)** откроется диалоговое окно **Изменение имени**, внешне идентичное окну **Создание имени**. В этом окне вы можете изменить любое свойство имени, за исключением области определения. При изменении имени диапазона все формулы рабочей книги, на него ссылающиеся, будут автоматически обновлены.



Совет

Для изменения области определения имя нужно вначале удалить, а затем воссоздать. Если при этом вы точно введете то же имя, все формулы, его использующие, останутся работоспособными.

Диалоговое окно **Изменение имени** является не единственным местом редактирования имен. Если единственное свойство, которое вы хотите изменить, это диапазон, вы можете это сделать непосредственно в главном окне диспетчера имен. В нижней части этого диалогового окна находится поле **Диапазон**. Выберите в списке нужное имя и измените ссылку в поле **Диапазон**.



Совет

Если вы редактируете содержимое поля **Диапазон** вручную, в строке состояния отображается слово *Укажите*. Если вы попытаетесь использовать при редактировании этого поля клавиши навигации, такие как стрелки, клавиши <Home> или <End>, то увидите, что перемещаетесь по рабочему листу, а не редактируете поле. Для переключения из режима указания в режим редактирования нужно нажать клавишу <F2> — в строке состояния теперь отображается слово *Правка*.

Удаление имен

Щелчок на кнопке **Удалить** удаляет выделенное имя из рабочей книги. Естественно, Excel предупредит вас о том, что данную операцию нельзя будет отменить. К сожалению, Excel не достаточно умен, чтобы заменить все ссылки на удаленное имя соответ-

вующими диапазонами ячеек. В результате все формулы, использующие удаленное имя, будут отображать ошибку #NAME?.

Способы присвоения имен ячейкам и диапазонам

В Excel существует несколько способов присвоения имен ячейкам и диапазонам, отличных от использования диспетчера имен. В данном разделе эти методы будут описаны. Кроме того, вы также найдете и другую полезную информацию, имеющую отношение к именам.

Присвоение имен в окне Присвоение имени

Для того чтобы присвоить имя ячейке или диапазону, сначала выделите нужную ячейку или диапазон. Затем выберите во вкладке **Формулы** команду **Присвоить имя** ⇒ **Присвоить имя (Defined Names) ⇒ Define Name** или нажмите комбинацию клавиш <Ctrl+F3>. Появится диалоговое окно, функционально и внешне идентичное окну **Создание имени диспетчера имен** (см. рис. 3.2).

Правила создания имен

Что касается назначаемых вами имен, Excel — достаточно неприхотливая программа. И все же существуют некоторые правила создания имен.

- Имена не могут содержать пробелов. Для разделения слов внутри имени можно использовать символ подчеркивания или точку (например, `Ежегодный_Итог` или `Ежегодный.Итог`).
- В качестве имени можно использовать любую комбинацию букв и цифр. Однако имя обязательно должно начинаться с буквы или символа подчеркивания. Имя не может начинаться с цифры (как, например, `3йКвартал`) или выглядеть как ссылка на ячейку (как, например, `К3`).
- Для разделения слов внутри имени используются только символы подчеркивания и точки. Этот момент не оговорен в документации к программе, однако Excel воспринимает и такие символы, как обратная косая черта (\) и вопросительный знак (?). Однако отмечу, что имя не может начинаться с этих символов.
- Количество символов имени не должно превышать 255. Поверьте моему опыту — использовать имена такой длины не имеет смысла. Применение чрезмерно длинных имен лишает всякого смысла операцию присвоения имени диапазону.
- В качестве имени вы можете использовать отдельные буквы (кроме R и C). Однако я бы не рекомендовал давать такие имена, поскольку имя должно нести информацию о содержимом ячейки или диапазона.
- При вводе имен регистр значения не имеет. `ЕжегодныйИтог` значит то же самое, что и `ежегодныйитог`. Excel сохраняет имя точно в таком же виде, в каком вы вводите его первоначально. Но, записывая имя в формуле, вы можете использовать любой регистр.

Excel использует несколько так называемых внутренних имен. Вы можете присваивать внутренние имена Excel ячейкам и диапазонам. Однако этого делать не рекомендуется. Итак, присваивая имя, вы должны избегать следующих имен: `Область_печати`, `Заголовок_печати`, `Область_консолидации`, `База_данных`, `Критерий`, `Фильтр` и `Название_листа`.



Примечание

Количество имен ячеек и диапазонов не ограничено. Excel позволяет присваивать одному и тому же элементу сразу несколько имен, однако делать этого не следует. Если выделенная ячейка или диапазон имеет несколько имен, в поле Имя всегда будет появляться первое из них по алфавиту.

Имя можно присвоить и несмежному диапазону ячеек. Выделить все ячейки и диапазоны несмежного диапазона можно с помощью мыши, удерживая клавишу <Ctrl>.

Присвоение имени с помощью поля Имя

В поле Имя вы сможете присвоить имя ячейке или диапазону значительно быстрее. Поле Имя — это раскрывающийся список, расположенный в левой части строки формул. Для того чтобы присвоить имя, выделите нужную ячейку или диапазон, затем щелкните в поле Имя и введите имя. Для подтверждения выбранного имени нажмите <Enter>. Если такое имя уже существует, то поле Имя нельзя использовать для его переназначения другому диапазону или ячейке. Если вы попытаетесь сделать это, программа выделит исходный диапазон. Назначить существующее имя другому диапазону или ячейке можно только с помощью диалогового окна Присвоение имени.



Предупреждение

Создавая имя в поле Имя, не забудьте нажать <Enter>. Если вы введете имя, а затем щелкнете в области рабочего листа, новое имя не будет создано.

Поле Имя выполняет еще одну функцию. С его помощью вы можете быстро активировать именованную ячейку или диапазон, как показано на рис. 3.3. Для того чтобы выделить ячейку или диапазон, щелкните в поле Имя и выберите нужное имя из списка. Будет выделена соответствующая ячейка или диапазон. Как ни странно, для активизации поля Имя не предусмотрена комбинация клавиш. Иными словами, вы не получите доступ к этому полю с помощью клавиатуры; сделать это можно только используя мышь. После того как вы щелкнете в поле Имя, для указания имени достаточно использовать клавиши со стрелками и <Enter>.




Рис. 3.3. Используя поле Имя, вы сможете быстро активизировать именованную ячейку или диапазон



Совет

Имена, созданные в поле Имя, по умолчанию имеют область определения уровня рабочей книги. Для того чтобы создать имя уровня рабочего листа, предварите его именем рабочего листа и восклицательным знаком (например, Лист2!Итог). Так как поле Имя работает только с текущим выделенным диапазоном, ввод имени листа, отличного от текущего, приведет к ошибке.

Автоматическое присвоение имен

Рабочий лист может содержать определенный текст, который вы хотите использовать в качестве имени смежных ячеек или диапазонов. На рис. 3.4 вы видите пример такого листа. В данном примере текст, находящийся в столбце А, может быть использован для создания имен ячеек с соответствующими значениями столбца В. Выполнить эту задачу в Excel достаточно просто.

	А	В	С	Д
3	Месяц	Продажи		
4	Январь	60 983,00р.		
5	Февраль	61 960,00р.		
6	Март	62 913,00р.		
7	Апрель	63 557,00р.		
8	Май	64 130,00р.		
9	Июнь	65 285,00р.		
10	Июль	66 449,00р.		
11	Август	67 302,00р.		
12	Сентябрь	68 065,00р.		
13	Октябрь	68 779,00р.		
14	Ноябрь	68 982,00р.		
15	Декабрь	70 789,00р.		
16				

Рис. 3.4. Excel позволяет создавать имена на основе текста, содержащегося в смежных ячейках

Для того чтобы присвоить имена, используя текст смежных ячеек, выделите данный текст и ячейки, которым должно быть присвоено имя (это могут быть как отдельные ячейки, так и диапазоны ячеек). Текст имен должен находиться в смежных ячейках относительно ячеек, которым вы хотите назначить имена (вы можете создать несколько выделений). Во вкладке **Формулы** выберите команду **Создать из выделенного фрагмента (Create from Selection)** или нажмите комбинацию клавиш <Ctrl+Shift+F3>. Откроется диалоговое окно, показанное на рис. 3.5.

Флажки в диалоговом окне расставлены на основе проведенного программой анализа выделенного диапазона. Например, если Excel обнаруживает текст в первой строке выделенного диапазона, программа предлагает создать имена в строке выше. Если Excel задала опции неправильно, то вы можете изменить предложенный вариант вручную. Щелкните на кнопке **ОК**, и Excel создаст имена. Обратите внимание на следующий факт: когда Excel создает имена, используя текст смежных ячеек, ячейки с текстом не включаются в именованный диапазон.

Если в результате использования текста смежной ячейки получится некорректное имя, Excel изменит его на правильное. Например, если ячейка содержит текст **Общий доход** (его использование в качестве имени некорректно, поскольку присутствует пробел), Excel преобразует пробел в символ подчеркивания и создаст имя **Общий_доход**.

Если же вместо текста смежная ячейка содержит значение или формулу, программа не сможет создать корректное имя. В таком случае Excel просто не создаст никакого имени.




Рис. 3.5. Диалоговое окно Создать имена



Предупреждение

Внимательно проверяйте имена, автоматически создаваемые программой. Иногда диалоговое окно Создание имен из выделенного диапазона работает противоречиво. На рис. 3.6 показана небольшая таблица с текстом и значениями. Выделите всю таблицу, выберите команду Создать из выделенного фрагмента и примите предложения программы (флажки в строке выше и в столбце слева). В результате имя `Товары` будет относиться не к диапазону `A2:A6`, как вы ожидали, а к диапазону `B2:C6`. Если верхняя левая ячейка выделенного диапазона содержит текст, и вы выберете опции в строке выше и в столбце слева, Excel использует это имя для всего набора данных, исключая верхнюю строку и левый столбец. Поэтому прежде чем принять предложения программы, предлагаемые командой, убедитесь, что этот выбор соответствует вашим замыслам.

	A	B	C	D
1	Товары	Количество	Цена	
2	Карандаши	12	2,15	
3	Бумага	30	12,05	
4	Маркеры	6	3,25	
5	Стиллеры	2	8,21	
6	Папки	24	0,34	
7				

Рис. 3.6. Создание имен на основе данных такой таблицы может привести к неожиданным результатам

Имена столбцов и строк

Иногда имеет смысл назначить имя целому столбцу или строке. Нередко рабочий лист используется для хранения информации, которую вы вводите время от времени. На рис. 3.7 приведен пример такого листа. Если вы создаете имя для данных столбца B, вам придется изменять ссылку на диапазон, которому назначено это имя, каждый раз, когда вы вводите новые данные. В таком случае целесообразно присвоить имя целому столбцу.

	А	В	С
1	Дата	Продажи	
2	январ.01	783	
3	январ.02	1023	
4	январ.03	3044	
5	январ.04	2044	
6	январ.05	3569	
7	январ.06		
8	январ.07		
9	январ.08		
10	январ.09		
11	январ.10		

Рис. 3.7. В этой таблице, содержащей данные о дневном объеме продаж, используется именованный диапазон, который состоит из целого столбца



Компакт-диск

Все приводимые в настоящей главе примеры вы найдете в файле `Chapter3_RUS.xlsx` на компакт-диске книги.

Столбец В вы можете назвать, например, ДневныеПродажи. Если этот диапазон расположен на рабочем листе Лист3, ссылка будет выглядеть следующим образом:

`=Лист3!$B:$B`

Назначив имя, вы можете использовать его в формуле. Следующая формула, например, возвращает сумму всех значений столбца В:

`=СУММ(ДневныеПродажи)`

Имена, созданные программой Excel

Excel создает и свои собственные имена. Например, если вы зададите область печати для листа, Excel создаст имя `Область_печати`. Если назначить повторяющиеся строки или столбцы, которые должны быть напечатаны, программа создаст имя `Заголовки_печати` на уровне рабочего листа. Когда вы выполняете запрос, который возвращает данные в рабочий лист, Excel назначает этим данным свои имена. Ряд надстроек, входящих в пакет Excel, создает скрытые имена (см. врезку “Скрытые имена”).

Вы вправе изменить ссылки на любые имена, автоматически создаваемые программой. Однако следует ясно представлять себе, к чему эти изменения приведут.

Скрытые имена

Некоторые макросы и надстройки Excel создают скрытые имена. Эти имена присутствуют в рабочей книге, но не появляются в диалоговом окне диспетчера имен и в поле Имя. Надстройка “Поиск решения”, например, создает целый ряд скрытых имен. Как правило, скрытые имена можно игнорировать, но они могут стать причиной возникновения некоторых проблем. Если вы копируете лист из одной рабочей книги в другую, вместе с листом копируются и скрытые имена. При этом скрытые имена могут образовывать ссылку, которую довольно трудно обнаружить.

К сожалению, работать с именами в Excel нелегко. Например, вы не сможете просмотреть полный список имен, присутствующих в рабочей книге. В диалоговом окне

диспетчера имен вы увидите только имена уровня листа, присутствующие в активном рабочем листе. В этом окне вы не найдете скрытых имен.

Если вам необходим эффективный инструмент для работы с именами, используйте утилиту *Name Lister*, которая является частью пакета Power Utility Pak. Эта утилита отображает полный список имен. Кроме того, Name Lister позволяет фильтровать имена несколькими способами. Вы можете просмотреть только имена уровня рабочего листа или же только связанные имена. Данная утилита поможет найти и удалить имена, присвоенные неправильному диапазону. Пробную версию пакета Power Utility Pak вы можете загрузить с сайта <http://j-walk.com/ss>.



Присвоение имен в нескольких листах

Имена ячейкам и диапазонам могут быть присвоены в нескольких рабочих листах с помощью “трехмерной” ссылки. Но вы не можете просто выделить диапазон и ввести имя в поле Имя. Для того чтобы назначить имя в нескольких листах, необходимо использовать диалоговое окно **Создание имени**. Формат такой ссылки будет выглядеть следующим образом:

ПервыйЛист!ПоследнийЛист!СсылкаНаДиапазон

На рис. 3.8 продемонстрирован пример, в котором имя (*КубДанных*), назначенное диапазону A1 : C3, распространяется на листы Лист1, Лист2 и Лист3.




Рис. 3.8. Присвоение имени в нескольких листах

Разумеется, вы можете просто ввести трехмерную ссылку на диапазон в поле **Диапазон**. Но, если вы хотите создать имя, отметив диапазон мышью, здесь могут возникнуть некоторые трудности. Даже если вы попытаетесь выделить диапазон, распространяющийся на несколько рабочих листов, Excel не сможет использовать адрес выделенного вами диапазона в диалоговом окне **Создание имени**.

Для того чтобы назначить имя *КубДанных* диапазону A1 : C3, который распространяется на листы Лист1, Лист2 и Лист3, следуйте приведенным ниже инструкциям.

1. Активизируйте Лист1.
2. Во вкладке **Формулы** выберите команду **Диспетчер имен** или нажмите комбинацию клавиш <Ctrl+F3>. В открывшемся диспетчере имен щелкните на кнопке **Создать**. Откроется диалоговое окно **Создание имени**.
3. Введите **КубДанных** в текстовом поле **Имя**.
4. Активизируйте поле **Диапазон** и нажмите клавишу , чтобы удалить ссылку на диапазон.
5. Выделите диапазон A1 : C3 на листе Лист1. В поле **Диапазон** появится следующая запись:
=Лист1!\$A\$1 : \$C\$3
6. Удерживая нажатой клавишу <Shift>, щелкните на вкладке Лист3. Вы заметите, что программа по непонятным причинам меняет ссылку на диапазон ссылкой на одну ячейку. На данном этапе запись будет выглядеть так:
= 'Лист1 : Лист3 ' ! \$A\$1
7. Выделите диапазон A1 : C3 на листе Лист1 еще раз. На этот раз в поле **Диапазон** будет содержаться следующая запись:
= 'Лист1 : Лист3 ' ! \$A\$1 : \$C\$3
8. Теперь, когда поле **Диапазон** содержит нужную ссылку, щелкните на кнопке **ОК**. Диалоговое окно **Создание имени** будет закрыто.

Задав имя, вы сможете использовать его в формулах. Например, следующая формула вернет сумму значений диапазона, который носит имя *КубДанных*.

=СУММ(КубДанных)



Примечание

Имена, распространяющиеся на несколько листов, не отображаются в поле **Имя** и в диалоговом окне **Переход** (это окно отображается при выборе команды **Найти и выделить** ⇒ **Перейти вкладки Главная**). Иными словами, Excel позволяет назначать такие имена, но вы не сможете автоматически выделить ячейки, к которым они относятся.

Если вы вставите новый лист в рабочую книгу, в которой используются имена, относящиеся к нескольким листам, имена будут распространяться на новый лист только в том случае, если этот лист находится между первым и последним листом, заданным в определении имени. В предыдущем примере рабочий лист, вставленный между Лист1 и Лист2, оказывается в диапазоне *КубДанных*. Лист, вставленный перед Лист1 или после Лист3, не включается в диапазон.

Если вы удалите первый или последний лист, включенный в определение имени, Excel автоматически изменит запись в поле **Диапазон**. В рассматриваемом нами примере

удаление Лист1 приведет к следующим изменениям в поле Диапазон имени КубДанных:

```
= ' Лист2 : Лист3 ' ! $A$1 : $C$3
```

Имена, распространяющиеся на множество листов, должны иметь область определения рабочей книги. Если они определены на уровне рабочего листа, они будут функционировать корректно, однако в диалоговом окне диспетчера имен будет отображаться ошибка.

Работа с именами диапазонов и ячеек

Создав имена диапазонов и ячеек, вы можете выполнять над ними целый ряд операций. В этом разделе вы узнаете о том, как выполнять основные операции над именами диапазонов и ячеек.

Создание списка имен

Если вы создали достаточно большое количество имен, может возникнуть необходимость зафиксировать, к какому диапазону относится каждое из них. В дальнейшем это поможет вам выявить ошибки или документировать свою работу.

Вероятно, вы захотите создать список всех имен (и соответствующих им адресов) рабочей книги. Для того чтобы создать такой список, сначала выделите ячейку в пустой области рабочего листа (если вы создадите список имен, состоящий из двух столбцов, то он будет отображаться поверх находящейся в ячейках информации), а затем выберите команду **Использовать в формуле**⇒**Вставить имена (Use in Formula**⇒**Paste)** (или нажмите клавишу <F3>). Откроется диалоговое окно **Вставка имени** (рис. 3.9), в котором будут перечислены все заданные имена. Для того чтобы вставить список имен, щелкните на кнопке **Все имена (Paste List)**.



Рис. 3.9. Диалоговое окно Вставка имени



Предупреждение

В список имен не входят имена уровня рабочего листа, кроме имен активного листа.

Список имен, вставленный в рабочий лист, состоит из двух столбцов. В первом столбце перечислены имена, а во втором — адреса соответствующих диапазонов. Адреса диапазонов, расположенные во втором столбце списка, представляют собой строки текста, которые напоминают формулы. Преобразовать такую строку в формулу можно, отредактировав ячейку (нажмите <F2>, а затем <Enter>). В результате строка становится формулой. Если имя относится к отдельной ячейке, формула отображает текущее значе-

ние этой ячейки. Если же имя относится к диапазону, формула возвращает ошибку #ЗНАЧ!.

Использование имен в формулах

Задав имя ячейки или диапазона, вы можете использовать его в формулах. Следующая формула, например, предназначена для вычисления суммы значений диапазона, который носит имя `ПроданоЕдиниц`:

```
=СУММ(ПроданоЕдиниц)
```

Когда имя, заданное на уровне рабочего листа, используется в формуле того листа, в котором оно было задано, имя рабочего листа не следует указывать перед именем диапазона. Если такое имя используется в другом листе рабочей книги, необходимо задать полное имя (имя рабочего листа, восклицательный знак и имя диапазона). Например, если `ПроданоЕдиниц` — имя, заданное на уровне рабочего листа `Лист1`, то формула для вычисления суммы диапазона на любом другом рабочем листе будет выглядеть так:

```
=СУММ(Лист1!ПроданоЕдиниц)
```

При вводе формулы вы можете выбрать команду **Вставка⇒Имя⇒Вставить** (или просто нажать клавишу <F3>). Откроется диалоговое окно **Вставка имени**. Выберите имя из списка, щелкните на кнопке **ОК**, и Excel вставит это имя в формулу. Как уже отмечалось, в списке диалогового окна **Вставка имени** представлены все имена уровня рабочей книги и имена уровня рабочего листа для активного листа.

Если в формуле вы использовали имя, которого не существует, Excel выдаст ошибку #ИМЯ?, тем самым сообщая вам, что не может найти имя, которое вы хотите использовать в формуле. Чаще всего появление такого сообщения свидетельствует о допущенной вами ошибке при написании имени.



Новинка

Ввод знака равенства в Excel 2007 активизирует функцию автозаполнения, как при вводе открывающей скобки в функции рабочего листа. При этом под активной ячейкой открывается список определенных имен и функций рабочего листа. Для использования функции автозаполнения вводите имя, пока оно не будет выделено в списке, а затем нажмите <Tab>, чтобы завершить ввод.

Если в формуле вы используете несуществующее имя, Excel отобразит ошибку #ИМЯ?, указывающую на то, что программа не может найти имя, которое вы пытаетесь использовать. Зачастую причиной такой ошибки является опечатка.

Использование операторов пересечения

В качестве оператора пересечения диапазонов в Excel служит единичный символ пробела. Например, следующая формула используется для вычисления суммы ячеек на пересечении двух диапазонов — `B1 : C20` и `A8 : D8`:

```
=СУММ(B1 : C20 A8 : D8)
```

Данные диапазоны пересекаются в ячейках `B8` и `C8`.

Оператор пересечения работает и с именованными диапазонами. На рис. 3.10 показана таблица, содержащая именованные диапазоны для столбцов и строк таблицы. Например, имя `Январь` относится к диапазону `B2 : E2`, а имя `Север` — к диапазону `B2 : B13`.

Следующая формула возвращает содержимое ячейки на пересечении диапазонов Январь и Север:

=Январь Север

	Север	Юг	Запад	Восток
Январь	235	465	118	325
Февраль	465	350	407	262
Март	256	361	355	381
Апрель	432	354	357	356
Май	495	474	414	386
Июнь	275	185	429	422
Июль	632	454	419	424
Август	457	545	499	459
Сентябрь	421	439	492	495
Октябрь	452	437	438	465
Ноябрь	652	552	494	466
Декабрь	571	613	562	538

Рис. 3.10. Эта таблица содержит именованные диапазоны, в которых заданы названия для столбцов и строк

Используя символ пробела для разделения двух ссылок на диапазоны, вы задаете так называемое *явное пересечение*. Это означает, что вы указываете Excel явно определить пересечение диапазонов. Программа же может находить и *неявные пересечения*. Неявное пересечение происходит в том случае, когда Excel выбирает значение из многоячеечного диапазона, основываясь на строке или столбце формулы, имеющей ссылку. Для того чтобы разобраться с пересечениями такого типа, рассмотрим конкретный пример. На рис. 3.11 показана таблица, содержащая именованный диапазон (B3:B8). В ячейке D5 отображается простая формула, приведенная ниже:

=МоиДанные

	A	B	C	D	E	F	G
1							
2							
3		6					
4		12					
5		18		18			
6		24					
7		30					
8							
9							
10							

Рис. 3.11. Диапазон B3:B8 этого рабочего листа имеет имя МоиДанные. Ячейка D5 демонстрирует неявное пересечение

Обратите внимание на то, что в ячейке D5 отображено значение из диапазона МоиДанные, которое соответствует строке формулы. Аналогично, если вы введете ту же формулу в строки с 3 по 8, формула представит соответствующее значение из диапазона

МоиДанные. Excel выполняет неявное пересечение, используя диапазон МоиДанные и строку, содержащую формулу. Это действие подобно вычислению следующей формулы:

```
=МоиДанные 5:5
```

Если ввести формулу в строку, не занятую диапазоном МоиДанные, то формула вернет ошибку, поскольку неявное пересечение в таком случае произойти не может.

Между прочим, неявные пересечения работают не только с именованными диапазонами. В рассматриваемом нами примере мы получим тот же результат, если в ячейке D5 будет находиться следующая формула (в которой не используется имя диапазона):

```
=$B$2:$B$8
```

Использование операторов диапазона с именами

Работая с именованными диапазонами, вы можете использовать оператор диапазона, в качестве которого в Excel применяется символ двоеточия (:). Вернемся к примеру на рис. 3.10. Следующая формула возвращает сумму значений ячеек столбцов от Север до Восток и строк от Январь до Март.

```
=СУММ((Север Январь):(Запад Март))
```

Ссылка на отдельную ячейку именованного диапазона

Для того чтобы вернуть значение ячейки диапазона, можно использовать функцию Excel ИНДЕКС. Предположим, определенный диапазон A1:A50 имеет название ДиапазонДанных. Следующая формула возвращает второе значение (значение ячейки A1) диапазона ДиапазонДанных.

```
=ИНДЕКС(ДиапазонДанных;2)
```

Второй и третий аргументы функции ИНДЕКС — необязательные, однако один из них всегда должен быть задан. Второй аргумент (используемый в приведенной формуле) обозначает смещение строки в диапазоне ДиапазонДанных.

Если диапазон ДиапазонДанных состоит из нескольких ячеек, расположенных в одной строке, используйте следующую формулу, в которой пропущен второй аргумент функции ИНДЕКС, но задан третий аргумент, определяющий смещение столбца в диапазоне ДиапазонДанных:

```
=ИНДЕКС(ДиапазонДанных; ;2)
```

Если же диапазон состоит из нескольких строк и столбцов, необходимо применить оба аргумента функции ИНДЕКС. Следующая формула, например, возвращает значение ячейки, расположенной в четвертой строке и пятом столбце диапазона ДиапазонДанных:

```
=ИНДЕКС(ДиапазонДанных;4;5)
```

Применение имен в уже существующих формулах

Если вы создали имя для ячейки или диапазона, Excel не будет автоматически заменять именами ссылки на эту ячейку или диапазон в формулах, созданных ранее. В то же время вы можете указать программе использовать новое имя в тех или иных формулах.

Для этого выделите диапазон, содержащий формулы, которые должны быть преобразованы. Затем выберите команду Присвоить имя⇒Применить имена (Define

Name⇒Apply Names). Откроется диалоговое окно **Применение имен**, показанное на рис. 3.12. В диалоговом окне **Применение имен** выберите те имена, которые вы хотите использовать в формулах. В формулах будут использованы только те имена, которые вы указали.




Рис. 3.12. Диалоговое окно **Применение имен**



Совет

Для того чтобы применить новое имя ко всем формулам рабочего листа, выделите одну ячейку перед тем, как выполнить команду **Применить имена**.

Флажок **Игнорировать тип ссылки (Ignore Relative/Absolute)** отвечает за то, как Excel заменяет адрес диапазона именем. Обычно имя ячейки или диапазона задается как абсолютная ссылка. Если флажок **Игнорировать тип ссылки** установлен, Excel применит имя только в том случае, если ссылка в формуле полностью соответствует типу диапазона. В большинстве случаев, когда вы применяете имена, удобнее игнорировать тип ссылки.

Если флажок **Использовать имена строк и столбцов (Use Row and Column Names)** включен, то при применении имен в формулах Excel использует оператор пересечения. Не имея возможности использовать точные имена ячеек, Excel для ссылок на ячейки использует имена строк и столбцов. Для того чтобы объединить имена, Excel использует оператор пересечения. Щелчком на кнопке **Параметры** можно задать несколько дополнительных параметров, но только при условии, что флажок **Использовать имена строк и столбцов** установлен.

Автоматическое применение имен при создании формул

Если вы вставляете в формулу ссылку на ячейку или диапазон, указывая этот элемент таблицы мышью, Excel автоматически заменяет адрес ячейки или диапазона именем (в том случае, если данному элементу присвоено имя).

Иногда это свойство программы очень помогает в работе. Но если вы захотите использовать адрес ячейки, а не ее имя, это свойство будет только мешать. К сожалению, отключить данную функцию невозможно. Если вы решили использовать адрес ячейки или диапазона, введите его вручную (не указывая элемент таблицы мышью).

Отмена применения имени

В Excel не предусмотрен специальный метод отмены применения имени. Иными словами, вы не сможете автоматически заменить имя ячейки или диапазона обычной ссылкой на этот элемент таблицы. Однако существует прием, с помощью которого можно заменить имя ссылкой. Для этого нужно установить режим вычислений по правилам Lotus 1-2-3. Откройте окно параметров программы и перейдите во вкладку Дополнительно. В группе Параметры совместимости с Lotus (Lotus Compatibility Settings) установите флажок Производить вычисления по правилам Lotus 1-2-3 (Transition Formula Entry), после чего щелкните на кнопке ОК.

Затем нажмите <F2> для того, чтобы отредактировать формулу, содержащую одно или несколько имен ячеек или диапазонов. Нажмите <Enter>, и редактирование ячеек будет завершено. Затем вернитесь в диалоговое окно параметров программы и снимите флажок Производить вычисления по правилам Lotus 1-2-3. Отредактированная таким образом ячейка вместо имен будет содержать относительные ссылки на диапазоны.



Примечание

Описанный выше прием не документирован и может работать не во всех случаях, так что внимательно проверяйте результат.

Удаление имен

Если вам не нужно ранее назначенное имя, то можете удалить его. При удалении имени диапазона содержимое этого диапазона остается без изменений. Выберите Вставка⇒Имя⇒Присвоить. Откроется диалоговое окно Присвоение имени. Укажите в списке имя, которое следует удалить, и щелкните на кнопке Удалить.



Предупреждение

Удаляя имена, будьте предельно внимательны. Если имя используется в формуле, его удаление повлечет за собой ошибку (в формуле появится сообщение об ошибке #ИМЯ?). Было бы очень удобно, если бы в формуле программа заменяла удаленные имена ссылками на ячейки и диапазоны. Но Excel, к сожалению, не делает этого. Обнаружив сообщение об ошибке #ИМЯ?, вы можете отменить удаление имени. Для того чтобы восстановить удаленное вами имя, выберите команду Правка⇒Отменить удаление.

Имена с ошибками

Если вы удалили строку или столбец, содержащий именованный диапазон или ячейку, само имя как таковое удалено не будет. Вместо этого в имени будет содержаться некорректная ссылка. К примеру, если ячейка A1 листа Лист1 имеет имя Дисконт, а вы удалили столбец A или строку 1, имя Дисконт будет содержать ссылку =Лист1!#ССЫЛ! (т.е. ошибочную ссылку). Если при этом имя Дисконт используется в какой-либо формуле, результатом формулы также будет #ССЫЛ!.

Чтобы избавиться от такой ошибки, нужно в диспетчере имен удалить это имя. Вы также можете переопределить имя так, чтобы оно указывало на корректную ячейку.

**Новинка**

Новый диспетчер имен программы Excel 2007 позволяет фильтровать отображаемые имена, используя predetermined фильтры. Один из этих фильтров специально предназначен для отображения только тех имен, которые имеют ошибочные ссылки. Это позволяет быстро локализовать проблемные имена.

Просмотр именованных диапазонов

Если вы измените масштаб изображения до 39% и меньше, вокруг именованных диапазонов увидите границы и имена, отображенные синим цветом (рис. 3.13). На печать эти границы и имена не выводятся — они предназначены только для того, чтобы вы могли видеть именованные диапазоны на экране.




Рис. 3.13. Если вы измените масштаб изображения до 39% и меньше, Excel отобразит имена диапазонов

**Примечание**

Этим приемом вы можете пользоваться только в Excel 97 и последующих версиях.

Использование имен в диаграммах

При создании диаграммы каждая последовательность данных имеет соответствующую формулу РЯД. Формула РЯД содержит ссылки на диапазоны, используемые в диаграмме. Если в диаграмме используется именованный диапазон, вы можете отредактировать формулу РЯД, заменив ссылку именем диапазона.

**Перекрестная ссылка**

Дополнительную информацию о диаграммах вы найдете в главе 17.

Поддержка имен ячеек и диапазонов

По мере того как вы редактируете или изменяете рабочий лист, Excel автоматически поддерживает созданные вами имена ячеек и диапазонов. В следующих примерах мы будем рассматривать рабочий лист Лист1, который содержит имя МойДиапазон, заданное на уровне рабочей книги, ссылающееся на диапазон из девяти ячеек:

=Лист1!\$C\$3:\$E\$5

Вставка строки или столбца

Когда вы вставляете строку выше именованного диапазона или столбец слева от именованного диапазона, Excel автоматически изменяет ссылку на этот диапазон так, чтобы она соответствовала новому адресу диапазона. Например, если вы вставите новую строку 1, имя `МойДиапазон` будет ссылаться на `=Лист1!C4:E6`.

Если вставить новую строку или столбец в именованный диапазон, этот диапазон расширится и будет включать в себя новые строки или столбцы. Например, если вы вставите новый столбец слева от столбца E, имя `МойДиапазон` будет относиться к `=Лист1!C3:F5`.

Удаление строки или столбца

Когда вы удаляете строку выше именованного диапазона или столбец слева от именованного диапазона, Excel автоматически изменяет ссылку на этот диапазон так, чтобы она соответствовала новому адресу диапазона. Например, если вы удалите строку 1, имя `МойДиапазон` будет относиться к `=Лист1!C2:E4`.

Если вы удалите строку или столбец внутри именованного диапазона, этот диапазон изменится соответствующим образом. Например, удалите столбец D, и имя `МойДиапазон` будет относиться к `=Лист1!C3:D5`.

Если вы удалите все строки и ячейки, составляющие именованный диапазон, именованный диапазон будет продолжать существовать, однако ссылка на него будет содержать ошибку. Например, удалите столбцы C, D, E, и имя `МойДиапазон` будет ссылаться на `=Лист1!#ССЫЛКА!`. Все формулы, в которых использовалось это имя, также будут возвращать ошибки.

Вырезание и вставка именованного диапазона

Когда вы вырезаете и вставляете именованный диапазон целиком, Excel изменяет ссылку соответствующим образом. Например, переместите диапазон `МойДиапазон` в другую часть листа, начинающегося с ячейки A1, и имя `МойДиапазон` будет сопоставлено ссылке `=Лист1!A1:C3`. Если вы вырежете и вставите только часть именованного диапазона, ссылка не изменится.

Потенциальные проблемы с именами

Имена необходимы для создания таблиц. Однако при работе с ними могут возникать некоторые проблемы. В этом разделе речь пойдет о тех моментах, о которых вы должны помнить, работая с именами.

Проблемы, возникающие при копировании рабочих листов

Как вы, вероятно, знаете, Excel позволяет копировать рабочие листы в пределах рабочей книги, а также в другую рабочую книгу. Прежде всего, мы рассмотрим пример копирования рабочего листа в пределах рабочей книги. Если копируемый лист содержит имена, заданные на уровне рабочего листа, те же имена будут присутствовать и в копии этого листа, причем в ссылках используется имя нового рабочего листа. Это именно то,

чего вы ожидали. Если же рабочая книга содержит имя, заданное на уровне рабочей книги, и данное имя относится к ячейке или диапазону копируемого рабочего листа, то оно также будет присутствовать и в копии. Однако это имя будет преобразовано в имя, заданное на уровне рабочего листа (чего вы, скорее всего, не ожидаете).

Рассмотрим рабочую книгу, содержащую один лист (Лист1). В этой книге присутствует имя — ИмяКниги, назначенное ячейке A1 на уровне рабочей книги. Кроме того, существует еще одно имя — Лист1!ЛокальноеИмя, назначенное ячейке A2 на уровне рабочего листа. Если вы сделаете копию листа Лист1 в пределах рабочей книги, новый лист будет называться Лист1 (2). Скопируйте лист, и вы обнаружите, что теперь рабочая книга содержит четыре имени (рис. 3.14).




Рис. 3.14. Копирование рабочего листа приводит к созданию дубликатов имен

Такое увеличение количества имен при копировании листа не только сбивает с толку, но и может привести к ошибкам, найти причину возникновения которых бывает непросто. Если в рассматриваемом нами примере мы введем следующую формулу в скопированный лист, формула вернет содержимое ячейки A1 скопированного листа:

=ИмяКниги

Иными словами, используется вновь созданное на уровне рабочего листа имя (а не исходное имя, заданное на уровне рабочей книги).

Если вы скопируете лист рабочей книги, которая содержит имя диапазона, относящегося к нескольким листам, вместе с этим рабочим листом будет скопировано и имя диапазона. В поле Диапазон появится сообщение об ошибке #ССЫЛКА!

Если же вы копируете лист в другую рабочую книгу, все имена исходной рабочей книги, относящиеся к ячейкам копируемого листа, будут скопированы в новую книгу. Это касается как имен, заданных на уровне рабочей книги, так и имен, заданных на уровне рабочего листа.



Примечание

Копирование и вставка ячеек из одного рабочего листа в другой не приводит к копированию имен даже в том случае, если копируемый диапазон содержит именованные ячейки.

Какой можно сделать вывод? Копируя листы рабочей книги, содержащей имена, будьте очень внимательны. Скопировав лист, проверьте все имена и удалите те, которые были скопированы случайно.

Проблемы, возникающие при удалении рабочих листов

При удалении листа, который содержит ячейки, используемые в заданном на уровне рабочей книги имени, имя не удаляется. Оно остается в рабочей книге, но ссылка на ячейку или диапазон в поле **Формула** будет содержать ошибку.

На рис. 3.15 показано диалоговое окно диспетчера имен, в котором присутствует имя с ошибкой. Исходная рабочая книга содержала лист Лист1 и ячейку A1 с именем ИмяКниги, заданным на уровне рабочей книги. После удаления листа Лист1 имя ИмяКниги осталось в рабочей книге. Однако в поле **Диапазон** диалогового окна **Присвоение имени** появилось следующее выражение:

```
=#ССЫЛКА!$A$1
```




Рис. 3.15. Удаление листа, содержащего диапазон МойДиапазон, приводит к появлению ошибочной ссылки

Имена объектов

Когда вы добавляете на рабочий лист объект, имя, присваиваемое ему по умолчанию, отражает тип объекта (например, **Прямоугольник1**).

Чтобы изменить имя объекта, выделите нужный объект, введите новое имя в поле **Имя** и нажмите <Enter>. Исключение составляют диаграммы. В этом случае необходимо выделить диаграмму, и в контекстной вкладке **Работа с диаграммами**⇒**Макет** (Chart Tools⇒Layout) выбрать команду **Свойства** (Properties).

Поле **Имя** в Excel имеет некоторые особенности: с его помощью можно переименовать объект, однако список объектов в этом окне не отображается. Более того, Excel позволяет задавать диапазонам имена, уже присвоенные объектам, при этом несколько объектов могут иметь одно и то же имя. В диалоговом окне диспетчера имен имена объектов не отображаются.

Обычно ошибочные имена в рабочей книге не вызывают каких-либо проблем, и все же рекомендуется удалять или корректировать все имена, содержащие ошибочные ссылки.

Ключ к пониманию термина “имя”

Пользователи Excel часто обращаются к *именованным ячейкам* и *именованным диапазонам*. В этой главе мы постоянно оперируем этими понятиями, однако данные термины не совсем точны.

Приведем ключ к пониманию термина *имя*. Присваивая имя, вы в действительности создаете именованную формулу. В отличие от обыкновенной формулы, именованная формула хранится не в ячейке, а в памяти Excel.

Если вы будете исходить из такой формулировки, вам будет легче овладеть сложными приемами, о которых пойдет речь в следующих разделах.

При работе в диалоговом окне диспетчера имен поле **Диапазон** содержит формулу, а поле **Имя** — ее имя. Вы уже заметили, что содержимое поля **Диапазон** всегда начинается со знака равенства, что свойственно формулам.

Как видно на рис. 3.16, рабочая книга содержит имя **Ставка**, присвоенное ячейке B1 листа **Лист1**. В поле **Диапазон** присутствует следующая запись:

```
=Лист1!$B$1
```

Когда вы используете имя **Ставка**, Excel в действительности вычисляет формулу, которая носит это имя, и возвращает результат. Например, вы можете ввести в ячейку следующую формулу:

```
=Ставка*1,05
```

Оценивая формулу, Excel, прежде всего, анализирует формулу, которая носит имя **Ставка** (оно существует только в памяти программы, а не в ячейке). Затем Excel умножает результат именованной формулы на 1,05 и возвращает результат. Разумеется, приведенная выше формула равноценна формуле, в которой вместо имени используется ссылка на ячейку:

```
=Лист1!$B$1*1,05
```

Возникает вопрос: можно ли создать такую именованную формулу, которая бы не содержала ссылок на ячейки? Ответ на него вы найдете в следующем разделе.

Присвоение имени константе

Рассмотрим таблицу, с помощью которой создается счет и вычисляется сумма налога для указанного объема продаж. Ставку налога с оборота вставьте в ячейку и ссылку на эту ячейку используйте в различных формулах. Для того чтобы было понятно, что за величина указана в ячейке, присвойте этой ячейке имя **СтавкаНалогов**.

Существует еще один способ. На рис. 3.17 представлен результат выполнения следующих инструкций.

1. Во вкладке **Формулы** выберите команду **Вставка⇒Имя⇒Присвоить имя (Define Name)**. Откроется диалоговое окно **Создание имени**.
2. Введите имя (в данном случае **СтавкаНалогов**) в поле **Имя**.
3. Перейдите в поле **Диапазон**, удалите содержащуюся в нем запись и вместо нее введите простую формулу, такую как **=0,075**.
4. Щелкните на кнопке **ОК** для того, чтобы закрыть диалоговое окно.



Рис. 3.16. Формально имя Ставка представляет собой именованную формулу, а не именованную ячейку



Рис. 3.17. Присвоение имени константе

В результате выполнения приведенных инструкций создается именованная формула, в которой не используются ссылки на ячейки. Чтобы проверить все вышеизложенное на практике, введем в любую ячейку такую формулу:

=СтавкаНалога

Эта простая формула возвращает 0,075 — результат именованной формулы СтавкаНалога. Поскольку именованная формула всегда возвращает один и тот же результат, ее можно считать именованной константой. Эта константа может быть использована и в более сложных формулах, как например:

=A1*СтавкаНалога

Если не изменять область определения с установленного по умолчанию на Лист, вы сможете использовать константу СтавкаНалога на всех листах рабочей книги.

Присвоение имени текстовым константам

В предыдущем примере константа представляла собой числовое значение. Однако константа может также состоять из текста. Например, вы можете определить константу для названия компании. Для создания следующей формулы, которая имеет название MS, используется диалоговое окно **Создание имени**:

= "Microsoft Corporation"

Теперь в любой ячейке можно использовать такую формулу, как:

= "Годовой отчет: "&MS

Данная формула вернет текст *Годовой отчет: Microsoft Corporation*.



Примечание

Имена, не относящиеся к диапазонам, не отображаются в поле **Имя** и в диалоговом окне **Переход** (которое можно вызвать нажатием клавиши <F5>). И это логично, поскольку константы отсутствуют в ячейках рабочего листа. Константы отображаются в диалоговом окне **Вставка имени**, и это тоже логично, поскольку имена констант используются в формулах.

Как вы, вероятно, догадываетесь, значение константы может быть изменено в любой момент в диалоговом окне **Создание имени**. Все, что нужно сделать, — это изменить значение в поле **Диапазон**. После того как диалоговое окно будет закрыто, Excel произведет пересчет всех формул, в которых фигурирует данная константа, с использованием нового значения.

Описанный выше прием очень эффективен во многих ситуациях, однако изменение значения занимает некоторое время. Если константу разместить в ячейке, изменить ее значение будет значительно легче. Если значение действительно является константой, изменять его вам не придется.

Использование функций рабочего листа в именованных формулах

На рис. 3.18 приведен еще один пример именованной формулы. В данном случае формула будет называться ТекущийМесяц и иметь следующий вид:

```
=МЕСЯЦ(СЕГОДНЯ())
```



Рис. 3.18. Создание именованной формулы, в которой используются функции рабочего листа

В формуле, показанной на рис. 3.16, были использованы две функции рабочего листа. Функция СЕГОДНЯ возвращает текущую дату, а функция МЕСЯЦ — номер месяца аргумента даты. Таким образом, вы можете ввести в ячейку следующую формулу, которая будет возвращать номер текущего месяца. Например, если текущий месяц — апрель, формула будет возвращать 4.

```
=ТекущийМесяц
```

Более эффективная именованная формула будет возвращать название месяца в текстовой форме. Такая формула, называемая ИмяМесяца, будет выглядеть следующим образом:

```
=ТЕКСТ(СЕГОДНЯ(); "MMMM")
```



Перекрестная ссылка

Детально функция ТЕКСТ() рассматривается в главе 5.

Теперь эту формулу можно ввести в ячейку, и она будет возвращать название текущего месяца в текстовой форме. В апреле формула будет возвращать текст Апрель.

```
=ИмяМесяца
```

Использование ссылок на ячейки и диапазоны в именованных формулах

На рис. 3.19 приведен еще один пример создания именованной формулы. На этот раз в формуле используется ссылка на ячейку. Эта формула, которая называется

ПервыйСимвол, возвращает первый символ, содержащийся в ячейке A1 листа Лист1. Именованная формула выглядит так:

=ЛЕВСИМВ(Лист1!\$A\$1;1)



Рис. 3.19. Создание именованной формулы, в которой используется ссылка на ячейку



Рис. 3.20. Создание именованной формулы, в которой используется ссылка на диапазон

Создайте такую именованную формулу, и вы сможете ввести ее в ячейку. Формула всегда возвращает первый символ, содержащийся в ячейке A1 листа Лист1.

=ПервыйСимвол

В следующей именованной формуле, которую мы рассмотрим, используется ссылка на диапазон. На рис. 3.20 показано диалоговое окно **Создание имени**, в котором задана именованная формула (под именем **Итого**).

=СУММ(Лист1!\$A\$1:\$D\$4)

Создав такую именованную формулу, вы сможете ввести ее в ячейку любого рабочего листа. Формула возвращает сумму значений ячеек диапазона A1:D4 листа Лист1.

=Итого

Обратите внимание на то, что в двух предыдущих формулах использовались абсолютные ссылки. По умолчанию все ссылки на ячейки и диапазоны, используемые в именованных формулах, являются абсолютными и включают название рабочего листа. Прочтите следующий раздел, и вы узнаете, какие интересные именованные формулы можно создавать, применяя относительные ссылки.

Использование именованных формул с относительными ссылками

Как уже отмечалось, при использовании диалогового окна **Создание имени** для создания именованной формулы со ссылками на ячейки или диапазоны, ссылки в поле **Диапазон** всегда абсолютны и включают название рабочего листа. В этом разделе речь пойдет об использовании в именованных формулах относительных ссылок на ячейки и диапазоны.

Использование относительной ссылки на ячейку

Начнем с простого примера. Для того чтобы создать именованную формулу с относительной ссылкой на ячейку, выполните следующее.

1. Откройте пустой рабочий лист.
2. Выделите ячейку A1 (этот шаг очень важен).

3. Во вкладке **Формулы** выберите команду **Присвоить имя**.
4. В поле **Имя** открывшегося окна введите **ЯчейкаСправа**.
5. Удалите содержимое поля **Диапазон** и введите следующую формулу (не указывая ячейку рабочего листа мышью):
`=Лист1!В1`
6. Щелкните на кнопке **ОК**, чтобы закрыть диалоговое окно **Создание имени**.
7. Введите в ячейку В1 необходимую информацию.
8. В ячейку А1 введите следующую формулу:
`=ЯчейкаСправа`
Вы обнаружите, что формула, расположенная в ячейке А1, возвращает содержимое ячейки В1.

Теперь скопируйте формулу ячейки А1 в несколько ячеек, расположенных под ней, и введите произвольные значения в столбец В. Вы увидите, что формула, находящаяся в столбце А, возвращает содержимое ячейки, расположенной справа. Иными словами, именованная формула **ЯчейкаСправа** имеет относительный характер.

Имя **ЯчейкаСправа** может быть использовано в любой ячейке (а не только в ячейках столбца А). Например, если ввести **ЯчейкаСправа** в ячейку D12, эта ячейка будет возвращать содержимое ячейки E12.

Для того чтобы убедиться в том, что в именованной формуле **ЯчейкаСправа** действительно используется относительная ссылка на ячейку, активизируйте любую ячейку, кроме А1, и откройте диалоговое окно **Присвоение имени** (рис. 3.21). В списке выберите **ЯчейкаСправа** и обратите внимание на содержимое поля **Формула**. Вы увидите, что изменение формулы зависит от активной в данный момент ячейки. Например, если активна ячейка E5, формула в диалоговом окне будет выглядеть так:
`=Лист1!F5`



Рис. 3.21. Формула под именем **ЯчейкаСправа** изменяется в зависимости от того, какая ячейка активна в данный момент

Если вы используете имя **ЯчейкаСправа** в другом рабочем листе, формула все равно будет ссылаться на ячейку, расположенную справа. И это будет ячейка с соответствующим адресом на листе **Лист1**. Дело в том, что именованная формула содержит ссылку на рабочий лист. Для того чтобы формула могла работать на любом рабочем листе, измените ее, следуя инструкциям.

1. Активизируйте ячейку A1 рабочего листа Лист1.
2. Во вкладке **Формулы** выберите команду **Диспетчер имен**.
3. В списке диалогового окна диспетчера имен выберите элемент **ЯчейкаСправа**.
4. Удалите содержимое поля **Диапазон** и введите следующую формулу:
=!B1
5. Щелкните на **ОК**, чтобы закрыть диспетчер имен.

Измените ссылку таким образом, и вы убедитесь, что теперь формула под именем **ЯчейкаСправа** корректно работает во всех листах рабочей книги.



Примечание

Именованная формула не будет работать, если вы используете ее в формуле, расположенной в столбце XFD, поскольку в этом случае формула ссылается на несуществующую ячейку (справа от столбца XFD столбцов больше нет).

Использование относительной ссылки на диапазон

Следующий пример основан на предыдущем. Речь пойдет о создании именованной формулы, которая возвращает сумму десяти ячеек, расположенных справа от определенной ячейки. Для того чтобы создать такую именованную формулу, следуйте приведенным ниже инструкциям.

1. Активизируйте ячейку A1.
2. Во вкладке **Формулы** выберите команду **Присвоить имя**.
3. В поле **Имя** открывшегося окна введите **Сумма10Ячеек**.
4. Введите следующую формулу в поле **Диапазон**:
=СУММ(!B1:!K1)

Создав именованную формулу, вы можете ввести следующую формулу в любую ячейку любого листа, и эта формула будет возвращать сумму десяти ячеек, расположенных справа:

=Сумма10Ячеек

Например, если вы введете такую формулу в ячейку D12, она будет возвращать сумму значений диапазона E12:N12, состоящего из десяти ячеек.

Обратите внимание на то, что, когда вы создавали именованную формулу, активной была ячейка A1, относительные ссылки в определении формулы заданы относительно ячейки A1. Заметьте также, что в формуле не использовано имя рабочего листа. Отсутствие имени рабочего листа (при условии, что в формулу включен восклицательный знак) говорит о том, что эту формулу можно использовать на любом листе.

Если вы активизируете ячейку D12, а затем откроете диалоговое окно **Создание имени**, то в поле **Диапазон** для формулы под именем **Сумма10Ячеек** увидите следующую запись:

=СУММ(!E12:!N12)



Примечание

Именованная формула `Сумма10Ячеек` не будет работать, если вы введете ее в ячейку, которая располагается в столбце, следующем за столбцом XET. В таком случае формула становится ошибочной, поскольку в ней появляется ссылка на несуществующую ячейку, расположенную за пределами столбца XFD.

Использование смешанных ссылок на диапазоны

Как уже отмечалось в главе 2, ссылка может быть абсолютной, относительной или смешанной. Смешанная ссылка на ячейку может состоять из следующих элементов.

- Абсолютной ссылки на столбец и относительной ссылки на строку (например, `$A1`).
- Относительной ссылки на столбец и абсолютной ссылки на строку (например, `A$1`).

Смешанные ссылки могут использоваться в именованных формулах. Применим такую ссылку на практике. Активизируйте ячейку B1. В диалоговом окне **Создание имени** создайте формулу под именем `ПерваяВСтроке`. Определение формулы должно выглядеть так:

```
=!$A1
```

В этой формуле использована абсолютная ссылка на столбец и относительная ссылка на строку. Следовательно, формула всегда возвращает значение, заданное в столбце A. Строка зависит от того, в какой ячейке вы введете формулу. Например, если вы введете следующую формулу в ячейку F12, эта формула вернет значение ячейки A12:

```
=ПерваяВСтроке
```



Примечание

Формулу `ПерваяВСтроке` нельзя использовать в столбце A, поскольку будет создана циклическая ссылка. Формула в данном случае будет ссылаться сама на себя.

Решение сложных задач с помощью формул

В этом разделе будет приведено несколько примеров решения усложненных задач с использованием имен. Все эти примеры предполагают, что вы уже знакомы с приемами, описанными в предыдущих разделах этой главы.

Функция `ДВССЫЛ` и именованный диапазон

Функция `ДВССЫЛ` позволяет определять адрес ячейки косвенно. Например, если ячейка A1 содержит текст "C45", формула будет возвращать содержимое ячейки C45:

```
=ДВССЫЛ(A1)
```

Функция `ДВССЫЛ` может использоваться и в работе с именованными диапазонами. На рис. 3.22 представлена таблица с 12 именами диапазонов, соответствующими названиям месяцев года. Например, имя `Январь` относится к диапазону B2:E2. Ячейка B16 содержит такую формулу:

```
=СУММ(ДВССЫЛ(A16))
```

На практике эта формула возвращает сумму именованного диапазона, который в текстовой форме задан в ячейке A16.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H
1		Север	Юг	Запад	Восток			
2	Январь	235	463	118	325			
3	Февраль	465	350	407	262			
4	Март	256	361	355	381			
5	Апрель	452	354	357	356			
6	Май	495	474	434	386			
7	Июнь	275	385	429	422			
8	Июль	632	454	419	434			
9	Август	457	545	499	459			
10	Сентябрь	421	439	492	495			
11	Октябрь	452	437	438	465			
12	Ноябрь	652	552	494	466			
13	Декабрь	573	611	582	538			
14								
15								
16	Март	=СУММ(ДВССЫЛ(A16))						
17								

Рис. 3.22. Использование функции ДВССЫЛ при работе с именованным диапазоном



Совет

Вы можете использовать команду Проверка данных (Data Validation) вкладки Данные для вставки в ячейку B16 раскрывающегося списка. (В поле Тип данных открывшегося окна выберите пункт Список, а в поле Источник введите диапазон A2:A13.) Теперь пользователь имеет возможность выбирать название месяца из списка, итоговое значение для которого будет отображаться в ячейке B16.

Используя функцию ДВССЫЛ, вы можете ссылаться и на имена уровня рабочего листа. Предположим, что вы работаете с рабочими листами Область1, Область2 и т.д. Каждый лист содержит имя ИтогиПродаж, заданное на уровне рабочего листа. Следующая формула извлекает значение из соответствующего рабочего листа, используя имя листа, заданное в ячейке A1:

=ДВССЫЛ(A1&"!ИтогиПродаж")

Использование функции ДВССЫЛ для создания именованного диапазона с фиксированным адресом

В Excel можно создать имя, которое бы всегда ссылалось на одну и ту же ячейку или диапазон даже в том случае, если вы вставите новые строки или столбцы. Предположим, вы хотите создать диапазон под именем ЛевыйВерхний, который бы всегда ссылался на ячейку A1. Если вы создадите имя обычным способом, вставка новой строки 1 приведет к тому, что диапазон ЛевыйВерхний будет ссылаться на ячейку A2. Вставка нового столбца приведет к тому, что диапазон ЛевыйВерхний будет ссылаться на ячейку B2. Для создания именованного диапазона, который имеет фиксированный адрес, необходимо задать именованную формулу, в поле Диапазон которой будет следующее выражение:

=ДВССЫЛ (" \$A\$1 ")

Если вы создадите такую формулу, диапазон ЛевыйВерхний будет ссылаться на ячейку A1 даже в том случае, если вы вставите новые строки или столбцы. Функция ДВССЫЛ предыдущей формулы позволяет задавать адрес ячейки косвенно с помощью текстового аргумента. Поскольку аргумент заключен в кавычки, он никогда не изменяется.



Примечание

В этой именованной формуле используется функция, потому формула не отображается ни в диалоговом окне Переход, ни в поле Имя.

Использование массивов в именованных формулах

Массив — это набор элементов. Массив может быть вертикальным и состоять из одного столбца, или горизонтальным и состоять из одной строки. Массивом может быть также совокупность нескольких столбцов и строк.



Перекрестная ссылка

В части IV настоящей книги подробно описаны массивы и формулы массивов. Однако эта тема имеет отношение и к именам.

Задать массив можно с помощью фигурных скобок. Каждый элемент массива отделяется запятой или точкой с запятой. Запятая используется для разделения элементов, расположенных вертикально, а точка с запятой — для разделения элементов, находящихся горизонтально.

В диалоговом окне **Создание имени** можно создать формулу ИменаМесяцев, которая имеет такое определение:

```
={"Янв"; "Фев"; "Мар"; "Апр"; "Май"; "Июн"; "Июл"; "Авг"; "Сен"; "Окт";  
"Ноя"; "Дек" }
```

Эта формула определяет массив, который состоит из 12 текстовых строк, расположенных горизонтально.



Примечание

Вводя такую формулу, не забудьте про фигурные скобки. В диалоговое окно **Создание имени** формула массива вводится не так, как в ячейку.

Определив формулу ИменаМесяцев, вы можете использовать ее в других формулах. Однако в конечной формуле необходимо обозначить, какой из элементов массива используется. Для этого рекомендуется использовать функцию ИНДЕКС. К примеру, следующая формула возвращает текст "Авг":

```
=ИНДЕКС (ИменаМесяцев ; 8)
```

Вы можете отобразить и весь массив, состоящий из 12 элементов, но для этого потребуется 12 смежных ячеек. Для того чтобы ввести 12 элементов массива в диапазон A1 : L1, следуйте приведенным ниже инструкциям.

1. В диалоговом окне **Создание имени** создайте формулу с именем ИменаМесяцев.
2. Выделите диапазон A1 : L1.
3. В строку формул введите =ИменаМесяцев.
4. Нажмите комбинацию клавиш <Ctrl+Shift+Enter>.

При нажатии <Ctrl+Shift+Enter> Excel вставляет формулу массива в выделенный диапазон. В данном случае формула будет введена в 12 смежных ячеек (рис. 3.23). Программа заключает формулу массива в скобки, напоминая вам, что это специфическая формула. В каждой из 12 ячеек диапазона A1:L1 формула будет иметь следующий вид:
{=ИменаМесяцев}




Рис. 3.23. Именованную формулу, которая содержит массив из 12 элементов, можно разместить в 12 смежных ячейках

Создание формулы с динамическим размером

Динамическая формула — это формула, относящаяся к диапазону, размер которого не фиксирован. Вероятно, осмыслить новое понятие непросто. Рассмотрим пример.

Обратите внимание на таблицу, показанную на рис. 3.24. Эта таблица содержит сведения об объеме продаж за период с января по май.

 A screenshot of an Excel spreadsheet window. The active cell is B6. The spreadsheet grid shows columns A through E and rows 1 through 15. The data is as follows:

	A	B	C	D	E
1	Январь	1 356,00р.			
2	Февраль	1 456,00р.			
3	Март	1 556,00р.			
4	Апрель	2 656,00р.			
5	Май	1 756,00р.			
6	Июнь				
7	Июль				
8	Август				
9	Сентябрь				
10	Октябрь				
11	Ноябрь				
12	Декабрь				
13					
14					
15					

Рис. 3.24. Динамическая формула может использоваться для ввода данных в столбец B

Предположим, вы хотите присвоить имя ОбъемПродаж данным, расположенным в столбце B. При этом имя не должно относиться к пустым ячейкам. Иными словами, ссылка на диапазон ОбъемПродаж будет меняться каждый месяц, по мере ввода новых данных. Разумеется, вы могли бы изменять ссылку вручную каждый месяц с помощью диалогового окна **Создание имени**, однако гораздо эффективнее использовать динамическую формулу, которая будет автоматически меняться с каждым последующим вводом данных.

Прежде чем задать динамическую формулу, создайте таблицу, показанную на рис. 3.24. Затем выполните следующие действия.

1. Вызовите диалоговое окно **Создание имени**.
2. В поле **Имя** введите **ОбъемПродаж**.
3. В поле **Диапазон** введите следующую формулу:
`=СМЕЩ(Лист1!B1;0;0;СЧЕТЗ(Лист1!$B:$B);1)`
4. Щелкните на кнопке **ОК**, чтобы закрыть диалоговое окно **Создание имени**.

В результате выполнения этих действий создается формула, в которой используются функции **СМЕЩ** и **СЧЕТЗ**. Для того чтобы проверить, как функционирует эта формула, введите следующую формулу в любую ячейку, которая не расположена в столбце В:

`=СУММ(ОбъемПродаж)`

Эта формула вернет сумму значений столбца В. Обратите внимание на то, что **ОбъемПродаж** не отображается ни в поле **Имя**, ни в диалоговом окне **Переход**. Для того чтобы выбрать диапазон, откройте диалоговое окно **Переход** и введите **ОбъемПродаж** вручную.

Напрашивается вопрос: зачем все эти трудности? Ведь и простая формула, приведенная ниже, сделает все, что необходимо:

`=СУМ(В:В)`

Эффективность работы именованных формул с динамическим размером становится очевидной при создании диаграмм. С помощью таких формул можно создать диаграмму с наборами данных, которые автоматически изменяются при вводе новых сведений.